



# INSIDE THE PROM

By Lance Rose

Accompanying this note are two assembly language programs for North Star's double density disk-controller PROM. The first program is a disassembled North Star PROM commented by Lance Rose; the second includes Rose's modifications for I/O port addressing, again heavily commented. A third assembly language program, a DOS modified by Rose

for I/O port addressing, is too long to print in Compass, but will be included on the INSUA library disk for Volume II Number 1 (the present issue) of Compass. Ask for INSUA disk Number 1010.

Readers interested in I/O port addressing should consult **DISKS ON N\* I/O PORT** in the previous issue of Compass (Volume I Number 4), by the same author.

;NORTH STAR DOUBLE DENSITY DISK CONTROLLER PROM

;  
;

```

      ORG      0E800H
LE800: MVI      C,0AH          ;Set retry counter to 10
      LDA      0EB15H        ;Turn on motors
      MVI      D,30H        ;Wait 48 sector times
      LXI      H,LE80D
      JMP      LE8D3
LE80D: LDA      0EA01H        ;Select drive 1
      LXI      H,LE816        ;Wait 2 sector times
      JMP      LE84D
LE816: MVI      B,0CH        ;Allow 12 sector times to find index hole
LE818: LXI      H,LE81E        ;Wait for sector pulse
      JMP      LE8D1
LE81E: LDA      0EB10H        ;Read A-status
      ANI      40H          ;Check for index hole detect
      JNZ      LE82D        ;Found it
      DCR      B            ;No, try again
      JNZ      LE818
LE82A: JMP      LE82A        ;Didn't find index hole, stop
LE82D: LDA      0EA21H        ;Set step direction to in
      LDA      0EA31H        ;Step in
      LDA      0EA21H
      JMP      LE84A
```

```

LE839:  LDA      0EB20H      ;Read B-status
        ANI      01H        ;Track zero?
        JNZ      LE852      ;Yes
        LDA      0EA01H      ;No, set step direction to out
        LDA      0EA11H      ;Step out
        LDA      0EA01H
LE84A:  LXI      H,LE839    ;Return to check for track zero
LE84D:  MVI      D,02H      ;Wait 2 sector times
        JMP      LE8D3
LE852:  LXI      H,LE858    ;Wait 1 sector time
        JMP      LE8D1
LE858:  LDA      0EB35H      ;Read C-status
        ANI      0FH        ;Mask off sector counter
        CPI      04H        ;Sector 4?
        JNZ      LE852      ;No, try next sector
LE862:  LDA      0EB10H      ;Read A-status
        ANI      04H        ;PLL enabled?
        JZ       LE862      ;No, keep trying
        MVI      A,09H      ;Time delay
LE86C:  DCR      A
        JNZ      LE86C
        LDA      0EB10H      ;Read A-status
        ANI      20H        ;Diskette double density?
        JNZ      LE897      ;Yes
        LDA      0EA21H      ;Set step direction to in
        LDA      0EA31H      ;Step in to track 1
        LDA      0EA21H
        LXI      H,LE887    ;Wait 2 sector times
        JMP      LE84D
LE887:  LXI      H,LE88D    ;Wait for sector pulse
        JMP      LE8D1
LE88D:  LDA      0EB35H      ;Read C-status
        ANI      0FH        ;Mask off sector counter
        CPI      08H        ;Sector 8?
        JNZ      LE887      ;No, try next sector
LE897:  MVI      B,8CH      ;Set time limit to find sync character
        LXI      D,0EB40H   ;Prepare to read data
LE89C:  LDA      0EB10H      ;Read A-status
        RRC
        JC       LE8AE      ;Sync detected?
        DCR      B          ;Yes
        JNZ      LE89C      ;No, try again
LE8A7:  DCR      C          ;Decrement retry counter
        JNZ      LE82D
LE8AB:  JMP      LE8AB      ;Tried 10 times, stop
LE8AE:  LDAX     D          ;Read first byte in block
        MOV      H,A        ;Use it for load address
        MVI      L,01H
        MOV      M,A
        RLC
        MOV      B,A
LE8B5:  LDAX     D          ;Read a byte
        MOV      M,A        ;Store it
        XRA      B          ;Compute CRC
        RLC
        MOV      B,A
        INR      L          ;Increment load address
        JNZ      LE8B5      ;If first 256 bytes, continue
        INR      H          ;Go to next 256 bytes

```

```

LE8BF: LDAX    D           ;Read a byte
      MOV    M,A         ;Store it
      XRA   B           ;Compute CRC
      RLC
      MOV    B,A
      INR   L           ;Increment load address
      JNZ   LE8BF       ;If more, load it
      LDAX  D           ;Read CRC check byte
      XRA   B           ;Compute CRC
      JNZ   LE8A7       ;If non-zero, retry
      DCR   H           ;Back up 100H for go address
      MVI   L,0AH       ;Jump to load address plus 10
      PCHL
LE8D1: MVI    D,01H      ;Wait for sector pulse
LE8D3: LDA    0EB11H     ;Reset sector flag
LE8D6: LDA    0EB10H     ;Read A-status
      ORA   A           ;Sector hole detected?
      JP    LE8D6       ;No, loop again
      DCR   D           ;Count # of sectors in D register
      LDA   0EB11H     ;Reset sector flag in case done
      JNZ   LE8D3       ;More sectors to go
      PCHL
      ;Done, return to address in H,L
      DB    00H,00H,00H,00H,00H,00H,00H,00H,00H,00H
      DB    00H,00H,00H,00H,00H,00H,00H,00H,00H,00H
      DB    00H,00H,00H,00H,00H,0E1H,0E9H
      END

```



```

;      North Star double density disk controller prom
;      modified for I/O port addressing
;
LE800 EQU    0E800H      ;Disk controller address
;
      ORG    0E800H
;
LE800: MVI    C,0AH      ;Set retry counter to 10
      MVI    A,15H      ;Turn on motors
      IN     LE800/100H+03H
      MVI    D,30H      ;Wait 48 sector times
      LXI   H,LE80D
      JMP    LE8D3
LE80D: MVI    A,01H      ;Select drive 1
      IN     LE800/100H+02H
      LXI   H,LE816      ;Wait 2 sector times
      JMP    LE84D
LE816: MVI    B,0CH      ;Allow 12 sector times to find index hole
LE818: LXI   H,LE81E      ;Wait for sector pulse
      JMP    LE8D1
LE81E: MVI    A,10H      ;Read A-status
      IN     LE800/100H+03H
      ANI   40H          ;Check for index hole detect
      JNZ   LE82D        ;Found it
      DCR   B           ;No, try again
      JNZ   LE818
LE82A: JMP    LE82A      ;Didn't find index hole, stop

```

```

LE82D:  MVI      A,21H           ;Set step direction to in
        IN       LE800/100H+02H
        MVI      A,31H           ;Step in
        IN       LE800/100H+02H
        MVI      A,21H
        IN       LE800/100H+02H
        JMP      LE84A
LE839:  MVI      A,20H           ;Read B-status
        IN       LE800/100H+03H
        ANI      01H             ;Track zero?
        JNZ      LE852           ;Yes
        MVI      A,01H           ;No, set step direction to out
        IN       LE800/100H+02H
        MVI      A,11H           ;Step out
        IN       LE800/100H+02H
        MVI      A,01H
        IN       LE800/100H+02H
LE84A:  LXI      H,LE839         ;Return to check for track zero
LE84D:  MVI      D,02H           ;Wait 2 sector times
        JMP      LE8D3
LE852:  LXI      H,LE858         ;Wait 1 sector time
        JMP      LE8D1
LE858:  MVI      A,35H           ;Read C-status
        IN       LE800/100H+03H
        ANI      0FH             ;Mask off sector counter
        CPI      04H             ;Sector 4?
        JNZ      LE852           ;No, try next sector
LE862:  MVI      A,10H           ;Read A-status
        IN       LE800/100H+03H
        ANI      04H             ;PLL enabled?
        JZ       LE862           ;No, keep trying
        MVI      A,09H           ;Time delay
LE86C:  DCR      A
        JNZ      LE86C
        MVI      A,10H           ;Read A-status
        IN       LE800/100H+03H
        ANI      20H             ;Diskette double density?
        JNZ      LE897           ;Yes
        MVI      A,21H           ;Set step direction to in
        IN       LE800/100H+02H
        MVI      A,31H           ;Step in to track 1
        IN       LE800/100H+02H
        MVI      A,21H
        IN       LE800/100H+02H
        LXI      H,LE887         ;Wait 2 sector times
        JMP      LE84D
LE887:  LXI      H,LE88D         ;Wait for sector pulse
        JMP      LE8D1
LE88D:  MVI      A,35H           ;Read C-status
        IN       LE800/100H+03H
        ANI      0FH             ;Mask off sector counter
        CPI      08H             ;Sector 8?
        JNZ      LE887           ;No, try next sector
LE897:  MVI      B,8CH           ;Set time limit to find sync character
LE89C:  MVI      A,10H           ;Read A-status
        IN       LE800/100H+03H
        RRC                    ;Sync detected?
        JC       LE8AE           ;Yes
        DCR      B               ;No, try again
        JNZ      LE89C

```

# LETS MEET

AGENDA FOR INSUA ANNUAL MEETING  
SAN FRANCISCO COMPUTER FAIRE

SAN FRANCISCO CIVIC AUDITORIUM

ROOM 403

MARCH 20, 1982 12 NOON TO 4 P.M.

12-1 pm Reception

1-3 pm Panel Discussion  
Tentative Panel:  
T. Warschauer  
Harley Licht

3 pm Formal Business  
Meeting including  
Elections, followed by  
informal discussion

4 pm Vacate Room

```
LE8A7: DCR      C                ;Decrement retry counter
        JNZ      LE82D
LE8AB:  JMP      LE8AB          ;Tried 10 times, stop
LE8AE:  MVI      A,40H         ;Read first byte in block
        IN       LE800/100H+03H
        MOV      H,A          ;Use it for load address
        MVI      L,01H
        MOV      M,A
        RLC                ;Start CRC check
        MOV      B,A
LE8B5:  MVI      A,40H         ;Read a byte
        IN       LE800/100H+03H
        MOV      M,A          ;Store it
        XRA      B            ;Compute CRC
        RLC
        MOV      B,A
        INR      L            ;Increment load address
        JNZ      LE8B5        ;If first 256 bytes, continue
        INR      H            ;Go to next 256 bytes
LE8BF:  MVI      A,40H         ;Read a byte
        IN       LE800/100H+03H
        MOV      M,A          ;Store it
        XRA      B            ;Compute CRC
        RLC
        MOV      B,A
        INR      L            ;Increment load address
        JNZ      LE8BF        ;If more, load it
        MVI      A,40H         ;Read CRC check byte
        IN       LE800/100H+03H
        XRA      B            ;Compute CRC
        JNZ      LE8A7        ;If non-zero, retry
        DCR      H            ;Back up 100H for go address
        MVI      L,0AH        ;Jump to load address plus 10
        PCHL                ;Execute
LE8D1:  MVI      D,01H        ;Wait for sector pulse
LE8D3:  MVI      A,11H        ;Reset sector flag
        IN       LE800/100H+03H
LE8D6:  MVI      A,10H        ;Read A-status
        IN       LE800/100H+03H
        ORA      A            ;Sector hole detected?
        JP       LE8D6        ;No, loop again
        DCR      D            ;Count # of sectors in D register
        MVI      A,11H        ;Reset sector flag in case done
        IN       LE800/100H+03H
        JNZ      LE8D3        ;More sectors to go
        PCHL                ;Done, return to address in H,L
        END
```